
Przyk≥ady programów

Interpreter prostego jÍzyka imperatywnego

I JÍzyk Imperator1 jest prostym jÍzykiem imperatywnym.
I Jego sk≥adniÍ opisuje poniøsza gramatyka BNF:

PROGRAM ::=
PROGRAM ::= INSTRUKCJA ; PROGRAM

INSTRUKCJA ::= IDENTYFIKATOR := WYRAØENIE
INSTRUKCJA ::= read IDENTYFIKATOR
INSTRUKCJA ::= write WYRAØENIE
INSTRUKCJA ::= if WARUNEK then PROGRAM fi
INSTRUKCJA ::= if WARUNEK then PROGRAM else PROGRAM fi
INSTRUKCJA ::= while WARUNEK do PROGRAM od

1NazwÍ zaproponowa≥ MKI.

WYRAØENIE ::= SK£ADNIK + WYRAØENIE
WYRAØENIE ::= SK£ADNIK - WYRAØENIE
WYRAØENIE ::= SK£ADNIK

SK£ADNIK ::= CZYNNIK * SK£ADNIK
SK£ADNIK ::= CZYNNIK / SK£ADNIK
SK£ADNIK ::= CZYNNIK mod SK£ADNIK
SK£ADNIK ::= CZYNNIK

CZYNNIK ::= IDENTYFIKATOR
CZYNNIK ::= LICZBA_NATURALNA
CZYNNIK ::= ( WYRAØENIE )

WARUNEK ::= KONIUNKCJA or WARUNEK
WARUNEK ::= KONIUNKCJA

KONIUNKCJA ::= PROSTY and KONIUNKCJA
KONIUNKCJA ::= PROSTY

PROSTY ::= WYRAØENIE = WYRAØENIE
PROSTY ::= WYRAØENIE /= WYRAØENIE
PROSTY ::= WYRAØENIE < WYRAØENIE
PROSTY ::= WYRAØENIE > WYRAØENIE
PROSTY ::= WYRAØENIE >= WYRAØENIE
PROSTY ::= WYRAØENIE =< WYRAØENIE
PROSTY ::= ( WARUNEK )

Example (Obliczenie sumy liczb A i B)

read A;
read B;
X := A;
Y := B;
while Y > 0 do

X := X + 1;
Y := Y - 1;

od;
write X;

I Program reprezentowany jest listπ z≥oøonπ z termów.
I Kaødy term reprezentuje jednπ instrukcjÍ.
I Jeúli instrukcja jest z≥oøona, to term zawiera jako podterm

listÍ termów reprezentujπcych zagnieødøone instrukcje.

PROGRAM = [ ]
PROGRAM = [INSTRUKCJA | PROGRAM]

INSTRUKCJA = assign(ID, WYRAØENIE)
INSTRUKCJA = read(ID)
INSTRUKCJA = write(WYRAØENIE)
INSTRUKCJA = if(WARUNEK, PROGRAM)
INSTRUKCJA = if(WARUNEK, PROGRAM, PROGRAM)
INSTRUKCJA = while(WARUNEK, PROGRAM)

Wyraøenie jest albo prostym wyraøeniem bÍdπcym zmiennπ lub
liczbπ naturalnπ albo wyraøeniem z≥oøonym bÍdπcym sumπ,
róønicπ, iloczynem lub ilorazem dwóch wyraøeÒ:

WYRAØENIE = id(ID)
WYRAØENIE = int(NUM)
WYRAØENIE = WYRAØENIE + WYRAØENIE
WYRAØENIE = WYRAØENIE - WYRAØENIE
WYRAØENIE = WYRAØENIE * WYRAØENIE
WYRAØENIE = WYRAØENIE / WYRAØENIE
WYRAØENIE = WYRAØENIE mod WYRAØENIE

Warunek jest relacjπ równoúci, róønoúci, mniejszoúci, wiÍkszoúci
(s≥abej lub silnej) miÍdzy wartoúciami dwóch wyraøeÒ albo
alternatywπ lub koniunkcjπ dwóch warunków:

WARUNEK = WYRAØENIE =:= WYRAØENIE
WARUNEK = WYRAØENIE =\= WYRAØENIE
WARUNEK = WYRAØENIE < WYRAØENIE
WARUNEK = WYRAØENIE > WYRAØENIE
WARUNEK = WYRAØENIE =< WYRAØENIE
WARUNEK = WYRAØENIE >= WYRAØENIE
WARUNEK = WARUNEK ; WARUNEK
WARUNEK = WARUNEK , WARUNEK

Example (Sumowanie liczb od 1 do N)

program1([
read(’N’),
assign(’SUM’, int(0)),
while(id(’N’) > int(0),

[assign(’SUM’, id(’SUM’) + id(’N’)),
assign(’N’, id(’N’) - int(1))]),

write(id(’SUM’))]).

Example (Obliczenie ilorazu D i reszty R z dzielenia M przez N)

program2([
read(’M’),
read(’N’),
assign(’D’, int(0)),
assign(’R’, id(’M’)),
while(id(’R’) >= id(’N’),

[assign(’D’, id(’D’) + int(1)),
assign(’R’, id(’R’) - id(’N’))]),

write(id(’D’)),
write(id(’R’))]).

Interpreter prostego jÍzyka imperatywnego
I Bieøπcy stan obliczeÒ zapisywaÊ bÍdziemy w postaci listy

asocjacji.
I Kaøda asocjacja jest termem w postaci ID = WartoúÊ, gdzie

ID jest nazwπ zmiennej a WartoúÊ jest bieøπcπ wartoúciπ tej
zmiennej.

% podstaw(+Stare, +ID, +WartoúÊ, -Nowe)
podstaw([], ID, N, [ID = N]).
podstaw([ID=_ | AS], ID, N, [ID=N | AS]) :- !.
podstaw([ID1=W1 | AS1], ID, N, [ID1=W1 | AS2]) :-

podstaw(AS1, ID, N, AS2).

% pobierz(+Asocjacje, +ID, -WartoúÊ)
pobierz([ID=N | _], ID, N) :- !.
pobierz([_ | AS], ID, N) :-

pobierz(AS, ID, N).

% wartoúÊ(+Wyraøenie, +Asocjacje, -WartoúÊ)
wartoúÊ(int(N), _, N).
wartoúÊ(id(ID), AS, N) :-

pobierz(AS, ID, N).
wartoúÊ(W1 + W2, AS, N) :-

wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N is N1 + N2.

wartoúÊ(W1 - W2, AS, N) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N is N1 - N2.

wartoúÊ(W1 * W2, AS, N) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N is N1 * N2.

wartoúÊ(W1 / W2, AS, N) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N2 =\= 0, N is N1 div N2.

wartoúÊ(W1 mod W2, AS, N) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N2 =\= 0,
N is N1 mod N2.

% prawda(+Warunek, +Asocjacje)
prawda(W1 =:= W2, AS) :-

wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N1 =:= N2.

prawda(W1 =\= W2, AS) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N1 =\= N2.

prawda(W1 < W2, AS) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N1 < N2.

prawda(W1 > W2, AS) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N1 > N2.

prawda(W1 >= W2, AS) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N1 >= N2.

prawda(W1 =< W2, AS) :-
wartoúÊ(W1, AS, N1), wartoúÊ(W2, AS, N2),
N1 =< N2.

prawda((W1, W2), AS) :-
prawda(W1, AS),
prawda(W2, AS).

prawda((W1; W2), AS) :-
( prawda(W1, AS),

!
; prawda(W2, AS)).

% interpreter(+Program, +Asocjacje)
interpreter([], _).
interpreter([read(ID) | PGM], ASSOC) :- !,

read(N),
integer(N),
podstaw(ASSOC, ID, N, ASSOC1),
interpreter(PGM, ASSOC1).

interpreter([write(W) | PGM], ASSOC) :- !,
wartoúÊ(W, ASSOC, WART),
write(WART), nl,
interpreter(PGM, ASSOC).

interpreter([assign(ID, W) | PGM], ASSOC) :- !,
wartoúÊ(W, ASSOC, WAR),
podstaw(AS, ID, WAR, ASSOC1),
interpreter(PGM, ASSOC1).

interpreter([if(C, P) | PGM], ASSOC) :- !,
interpreter([if(C, P, []) | PGM], ASSOC).

interpreter([if(C, P1, P2) | PGM], ASSOC) :- !,
( prawda(C, ASSOC)
-> append(P1, PGM, DALEJ)
; append(P2, PGM, DALEJ)),
interpreter(DALEJ, ASSOC).

interpreter([while(C, P) | PGM], ASSOC) :- !,
append(P, [while(C, P)], DALEJ),
interpreter([if(C, DALEJ) | PGM], ASSOC).

% interpreter(+Program)
interpreter(PROGRAM) :-

interpreter(PROGRAM, []).

Example (Sumowanie)

?- program1(X), interpreter(X).
|: 10.
55
X = [read(’N’), assign(’SUM’, int(0)),

while(id(’N’)>int(0), [assign(’SUM’, id(’SUM’)
+id(’N’)), assign(’N’, id(’N’)-int(1))]),
write(id(’SUM’))].

Example (Iloraz i reszta)

?- program2(X), interpreter(X).
|: 123.
|: 13.
9
6
X = [read(’M’), read(’N’), assign(’D’, int(0)),

assign(’R’, id(’M’)), while(id(’R’)>=id(’N’),
[assign(’D’, id(...)+int(...)), assign(’R’, ...
- ...)]), write(id(’D’)), write(id(’R’))].

?- X is 9*13+6.
X = 123.

Interpreter podzbioru Prologu w Prologu

I Zak≥adamy, øe predykaty zapisane sπ w postaci klauzul,
których cia≥a zawierajπ tylko koniunkcje formu≥ atomowych
(nie ma negacji, alternatyw i implikacji).

I Wszystkie predykaty sπ zdefiniowane przez nas (nie
korzystamy z predykatów wbudowanych, które mogπ byÊ
napisane np. w jÍzyku C).

Example (Konkatenacja list na liúcie)

app([], X, X).
app([X | L1], L2, [X | L3]) :-

app(L1, L2, L3).

app([], []).
app([L1 | L2], L3) :-

app(L1, L4, L3),
app(L2, L4).

Interpreter podzbioru Prologu w Prologu

I Predykat clause(H, B) dostarcza g≥owÍ H i cia≥o B klauzuli.
I H jest unifikowane z g≥owπ a B z cia≥em klauzuli.
I Fakt ma cia≥o równe true.

?- clause(app([], A, B), C).
A = B,
C = true.

?- clause(app(A, B, C), D).
A = [],
B = C,
D = true ;
A = [_G4888|_G4889],
C = [_G4888|_G4892],
D = app(_G4889, B, _G4892).

?- clause(app(A, B), C).
A = B, B = [],
C = true ;
A = [_G4870|_G4871],
C = (app(_G4870, _G4874, B), app(_G4871, _G4874)).

Interpreter podzbioru Prologu w Prologu

I Interpreter Prologu w Prologu zapiszemy w postaci predykatu
udowodnij(Cel), gdzie Cel jest zadanym celem do
udowodnienia.

I Celem moøe byÊ warunek true, formu≥a atomowa lub
koniunkcja dwóch celów.

udowodnij(true) :- !.
udowodnij((G1, G2)) :- !,

udowodnij(G1),
udowodnij(G2).

udowodnij(A) :-
clause(A, B),
udowodnij(B).

?- udowodnij(app(X, Y, [1, 2, 3])).
X = [],
Y = [1, 2, 3] ;
X = [1],
Y = [2, 3] ;
X = [1, 2],
Y = [3] ;
X = [1, 2, 3],
Y = [] ;
false.

?- udowodnij(app([1, 2, 3], X, [1, 2, 3, 4 | Y])).
X = [4|Y].

?- udowodnij(app([[1, 2], [3], [4, 5]], X)).
X = [1, 2, 3, 4, 5].

Unifikacja i reprezentacja termów
Idea algorytmu unifikacji termów:
unify(V, T ) :- var(V), !, bind(V, T).
unify(T, V ) :- var(V), !, bind(V, T).
unify(T1, T2) :- T1 =.. [Fun | Args1],

T2 =.. [Fun | Args2],
unify_args(Args1, Args2).

unify_args([], []).
unify_args([T1 | A1], [T2 | A2]) :-

unify(T1, T2),
unify_args(A1, A2).

bind(V, T) :- V = T.

?- unify(f(A, g(A)), f(a, B)).
A = a, B = g(a).

Reprezentacja termu w WAM (ang. Warren Abstract Machine):

str 10
f/21

a/010

ref 22
str 43

g/14
ref 25
str 76

f/27
str 108
ref 99

f(

A,
g(

A))
f(

B)
a,

str A’A

ref A’A

fun/nA

pod adresem A jest referencja do struktury pamiętanej pod adresem A’

pod adresem A jest referencja do adresu A’

pod adresem A jest n-argumentowy funktor fun

Dwa termy f(A, g(A)) i f(a, B) zajmują 11 komórek pamięci:

dref(A, X) do jakiego miejsca X prowadzi adres A?

?- dref(6, X).
X = 6.
?- dref(5, X).
X = 2.

bind(A, X) niech referencja pod adresem A 
prowadzi do miejsca X

Example (Unifikacja w WAM)

Reprezentacja komórek pamiÍci w postaci dynamicznych faktów:

:- dynamic store/2.

store(0, str(1)). % f(
store(1, f/2). %
store(2, ref(2)). % A,
store(3, str(4)). % g(
store(4, g/1). %
store(5, ref(2)). % A))
store(6, str(7)). % f(
store(7, f/2). %
store(8, str(10)). % a,
store(9, ref(9)). % B)
store(10, a/0). %

Example (Unifikacja w WAM cd.)

Pomocnicze predykaty:

dref(A, R) :-
store(A, ref(V)),
A =\= V, !,
dref(V, R).

dref(A, A).

bind(A1, A2) :-
retract(store(A1, _)),
assert(store(A1, ref(A2))).

Przyk≥ady programów

Unifikacja i reprezentacja termów

Example (Unifikacja w WAM cd.)

unify(A1, A2) :-
dref(A1, D1),
dref(A2, D2),
unify2(D1, D2).

unify2(D, D) :- !.
unify2(D1, D2) :-

store(D1, S1), % S1 = ref(A1) lub S1 = str(A1)
store(D2, S2), % S2 = ref(A2) lub S2 = str(A2)
S1 =.. [T1, A1],
S2 =.. [T2, A2],
unify3(D1, T1, A1, D2, T2, A2).

Example (Unifikacja w WAM cd.)

unify3(D1, ref, _, D2, _, _ ) :- !, bind(D1, D2).
unify3(D1, _, _, D2, ref, _ ) :- !, bind(D2, D1).
unify3(_, _, V1, _, _, V2) :- store(V1, F/N),

store(V2, F/N),
A1 is V1+1,
A2 is V2+1,
unify4(N, A1, A2).

unify4(0, _, _ ) :- !.
unify4(N, A1, A2) :- N1 is N-1,

unify(A1, A2),
NA1 is A1+1,
NA2 is A2+1,
unify4(N1, NA1, NA2).

Example (Unifikacja w WAM cd.)

Przyk≥adowe wywo≥anie:

?- unify(0, 6). % odpowiada f(A, g(A)) = f(a, B)
true.

Jaki jest wynik unifikacji? Jakie wartoúci przyjÍ≥y zmienne?

Example (Unifikacja w WAM cd.)

Zmiany zachodzπce w komórkach pamiÍci:

str 10
f/21

a/010

ref 22
str 43

g/14
ref 25
str 76

f/27
str 108
ref 99

f(

A,
g(

A))
f(

B)
a,

str 10
f/21

a/010

ref 82
str 43

g/14
ref 25
str 76

f/27
str 108
ref 99

f(

a,
g(

a))
f(

B)
a,

bind(2, 8)

str 10
f/21

a/010

ref 82
str 43

g/14
ref 25
str 76

f/27
str 108
ref 39

f(

a,
g(

a))
f(

g(a))
a,

bind(9, 3)

Example (Unifikacja w WAM cd.)

Wynik unifikacji:

f/2

g/1

f/2str 10
f/21

a/010

ref 82
str 43

g/14
ref 25
str 76

f/27
str 108
ref 39

A

B a/0

A B

A = a, B = g(a)


